

NAG Library Routine Document

F08TNF (ZHPGV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08TNF (ZHPGV) computes all the eigenvalues and, optionally, all the eigenvectors of a complex generalized Hermitian-definite eigenproblem, of the form

$$Az = \lambda Bz, \quad ABz = \lambda z \quad \text{or} \quad BAz = \lambda z,$$

where A and B are Hermitian, stored in packed format, and B is also positive definite.

2 Specification

```
SUBROUTINE F08TNF (ITYPE, JOBZ, UPLO, N, AP, BP, W, Z, LDZ, WORK, RWORK,      &
                  INFO)
INTEGER          ITYPE, N, LDZ, INFO
REAL (KIND=nag_wp) W(N), RWORK(3*N-2)
COMPLEX (KIND=nag_wp) AP(*), BP(*), Z(LDZ,*), WORK(2*N-1)
CHARACTER(1)     JOBZ, UPLO
```

The routine may be called by its LAPACK name *zhpgv*.

3 Description

F08TNF (ZHPGV) first performs a Cholesky factorization of the matrix B as $B = U^H U$, when $UPLO = 'U'$ or $B = LL^H$, when $UPLO = 'L'$. The generalized problem is then reduced to a standard symmetric eigenvalue problem

$$Cx = \lambda x,$$

which is solved for the eigenvalues and, optionally, the eigenvectors; the eigenvectors are then backtransformed to give the eigenvectors of the original problem.

For the problem $Az = \lambda Bz$, the eigenvectors are normalized so that the matrix of eigenvectors, Z , satisfies

$$Z^H AZ = \Lambda \quad \text{and} \quad Z^H BZ = I,$$

where Λ is the diagonal matrix whose diagonal elements are the eigenvalues. For the problem $ABz = \lambda z$ we correspondingly have

$$Z^{-1} A Z^{-H} = \Lambda \quad \text{and} \quad Z^H B Z = I,$$

and for $BAz = \lambda z$ we have

$$Z^H A Z = \Lambda \quad \text{and} \quad Z^H B^{-1} Z = I.$$

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: ITYPE – INTEGER *Input*
On entry: specifies the problem type to be solved.
 ITYPE = 1
 $Az = \lambda Bz.$
 ITYPE = 2
 $ABz = \lambda z.$
 ITYPE = 3
 $BAz = \lambda z.$
Constraint: ITYPE = 1, 2 or 3.
- 2: JOBZ – CHARACTER(1) *Input*
On entry: indicates whether eigenvectors are computed.
 JOBZ = 'N'
 Only eigenvalues are computed.
 JOBZ = 'V'
 Eigenvalues and eigenvectors are computed.
Constraint: JOBZ = 'N' or 'V'.
- 3: UPLO – CHARACTER(1) *Input*
On entry: if UPLO = 'U', the upper triangles of A and B are stored.
 If UPLO = 'L', the lower triangles of A and B are stored.
Constraint: UPLO = 'U' or 'L'.
- 4: N – INTEGER *Input*
On entry: n , the order of the matrices A and B .
Constraint: $N \geq 0$.
- 5: AP(*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array AP must be at least $\max(1, N \times (N + 1)/2)$.
On entry: the upper or lower triangle of the n by n Hermitian matrix A , packed by columns.
 More precisely,
 if UPLO = 'U', the upper triangle of A must be stored with element A_{ij} in $AP(i + j(j - 1)/2)$ for $i \leq j$;
 if UPLO = 'L', the lower triangle of A must be stored with element A_{ij} in $AP(i + (2n - j)(j - 1)/2)$ for $i \geq j$.
On exit: the contents of AP are destroyed.
- 6: BP(*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array BP must be at least $\max(1, N \times (N + 1)/2)$.
On entry: the upper or lower triangle of the n by n Hermitian matrix B , packed by columns.
 More precisely,

if $\text{UPLO} = \text{'U'}$, the upper triangle of B must be stored with element B_{ij} in $\text{BP}(i + j(j - 1)/2)$ for $i \leq j$;
 if $\text{UPLO} = \text{'L'}$, the lower triangle of B must be stored with element B_{ij} in $\text{BP}(i + (2n - j)(j - 1)/2)$ for $i \geq j$.

On exit: the triangular factor U or L from the Cholesky factorization $B = U^H U$ or $B = LL^H$, in the same storage format as B .

7: $W(N) - \text{REAL} (\text{KIND}=\text{nag_wp})$ array *Output*

On exit: the eigenvalues in ascending order.

8: $Z(LDZ, *) - \text{COMPLEX} (\text{KIND}=\text{nag_wp})$ array *Output*

Note: the second dimension of the array Z must be at least $\max(1, N)$ if $\text{JOBZ} = \text{'V'}$, and at least 1 otherwise.

On exit: if $\text{JOBZ} = \text{'V'}$, Z contains the matrix Z of eigenvectors. The eigenvectors are normalized as follows:

if $\text{ITYPE} = 1$ or 2 , $Z^H B Z = I$;

if $\text{ITYPE} = 3$, $Z^H B^{-1} Z = I$.

If $\text{JOBZ} = \text{'N'}$, Z is not referenced.

9: $LDZ - \text{INTEGER}$ *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08TNF (ZHPGV) is called.

Constraints:

if $\text{JOBZ} = \text{'V'}$, $LDZ \geq \max(1, N)$;
 otherwise $LDZ \geq 1$.

10: $\text{WORK}(2 \times N - 1) - \text{COMPLEX} (\text{KIND}=\text{nag_wp})$ array *Workspace*

11: $\text{RWORK}(3 \times N - 2) - \text{REAL} (\text{KIND}=\text{nag_wp})$ array *Workspace*

12: $\text{INFO} - \text{INTEGER}$ *Output*

On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$

F07GRF (ZPPTRF) or F08GNF (ZHPEV) returned an error code:

$\leq N$ if $\text{INFO} = i$, F08GNF (ZHPEV) failed to converge; i off-diagonal elements of an intermediate tridiagonal form did not converge to zero;

$> N$ if $\text{INFO} = N + i$, for $1 \leq i \leq N$, then the leading minor of order i of B is not positive definite. The factorization of B could not be completed and no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and vectors may be large, although when the diagonal elements of B differ widely in magnitude the eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section 4.10 of Anderson *et al.* (1999) for details of the error bounds.

The example program below illustrates the computation of approximate error bounds.

8 Parallelism and Performance

F08TNF (ZHPGV) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08TNF (ZHPGV) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is proportional to n^3 .

The real analogue of this routine is F08TAF (DSPGV).

10 Example

This example finds all the eigenvalues and eigenvectors of the generalized Hermitian eigenproblem $Az = \lambda Bz$, where

$$A = \begin{pmatrix} -7.36 & 0.77 - 0.43i & -0.64 - 0.92i & 3.01 - 6.97i \\ 0.77 + 0.43i & 3.49 & 2.19 + 4.45i & 1.90 + 3.73i \\ -0.64 + 0.92i & 2.19 - 4.45i & 0.12 & 2.88 - 3.17i \\ 3.01 + 6.97i & 1.90 - 3.73i & 2.88 + 3.17i & -2.54 \end{pmatrix}$$

and

$$B = \begin{pmatrix} 3.23 & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 \end{pmatrix},$$

together with an estimate of the condition number of B , and approximate error bounds for the computed eigenvalues and eigenvectors.

The example program for F08TQF (ZHPGVD) illustrates solving a generalized symmetric eigenproblem of the form $ABz = \lambda z$.

10.1 Program Text

```
Program f08tnfe

!     F08TNF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
Use nag_library, Only: f06udf, nag_wp, x02ajf, zhpgv, ztpcon
!     .. Implicit None Statement ..
Implicit None
!     .. Parameters ..
```

```

Integer, Parameter :: nin = 5, nout = 6
Character (1), Parameter :: uplo = 'U'
! .. Local Scalars ..
Real (Kind=nag_wp) :: anorm, bnorm, eps, rcond, rcondb,      &
                      t1, t2
Integer :: i, info, j, n
! .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: ap(:), bp(:), work(:)
Complex (Kind=nag_wp) :: dummy(1,1)
Real (Kind=nag_wp), Allocatable :: eerbnd(:), rwork(:), w(:)
! .. Intrinsic Procedures ..
Intrinsic :: abs
! .. Executable Statements ..
Write (nout,*) 'F08TNF Example Program Results'
Write (nout,*) Skip heading in data file
Read (nin,*) Read (nin,*) n
Read (nin,*) n

Allocate (ap((n*(n+1))/2),bp((n*(n+1))/2),work(2*n),eerbnd(n),rwork(3*n- &
2),w(n))

! Read the upper or lower triangular parts of the matrices A and
! B from data file

If (uplo=='U') Then
    Read (nin,*)((ap(i+(j*(j-1))/2),j=i,n),i=1,n)
    Read (nin,*)((bp(i+(j*(j-1))/2),j=i,n),i=1,n)
Else If (uplo=='L') Then
    Read (nin,*)((ap(i+((2*n-j)*(j-1))/2),j=1,i),i=1,n)
    Read (nin,*)((bp(i+((2*n-j)*(j-1))/2),j=1,i),i=1,n)
End If

! Compute the one-norms of the symmetric matrices A and B
anorm = f06udf('One norm',uplo,n,ap,rwork)
bnorm = f06udf('One norm',uplo,n,bp,rwork)

! Solve the generalized symmetric eigenvalue problem
! A*x = lambda*B*x (ITYPE = 1)

! The NAG name equivalent of zhpgv is f08tnf
Call zhpgv(l,'No vectors',uplo,n,ap,bp,w,dummy,1,work,rwork,info)

If (info==0) Then
    ! Print solution
    Write (nout,*) 'Eigenvalues'
    Write (nout,99999) w(1:n)

    ! Call ZTPCON (F07UU) to estimate the reciprocal condition
    ! number of the Cholesky factor of B. Note that:
    ! cond(B) = 1/RCOND**2

    Call ztpcon('One norm',uplo,'Non-unit',n,bp,rcond,work,rwork,info)

    ! Print the reciprocal condition number of B
    rcondb = rcond**2
    Write (nout,*) Write (nout,*) 'Estimate of reciprocal condition number for B'
    Write (nout,99998) rcondb

    ! Get the machine precision, EPS, and if RCOND is not less
    ! than EPS**2, compute error estimates for the eigenvalues

    eps = x02ajf()
    If (rcond>=eps) Then
        t1 = eps/rcondb
        t2 = anorm/bnorm

```

```

Do i = 1, n
    eerbnd(i) = t1*(t2+abs(w(i)))
End Do

!      Print the approximate error bounds for the eigenvalues

Write (nout,*)
Write (nout,*) 'Error estimates for the eigenvalues'
Write (nout,99998) eerbnd(1:n)
Else
    Write (nout,*)
    Write (nout,*) 'B is very ill-conditioned, error ', &
        'estimates have not been computed'
End If
Else If (info>n .And. info<=2*n) Then
    i = info - n
    Write (nout,99997) 'The leading minor of order ', i, &
        ' of B is not positive definite'
Else
    Write (nout,99996) 'Failure in ZHPGV. INFO =', info
End If

99999 Format (3X,(6F11.4))
99998 Format (4X,1P,6E11.1)
99997 Format (1X,A,I4,A)
99996 Format (1X,A,I4)
End Program f08tnfe

```

10.2 Program Data

F08TNF Example Program Data

```

4                                         :Value of N

(-7.36, 0.00) ( 0.77, -0.43) (-0.64, -0.92) ( 3.01, -6.97)
              ( 3.49, 0.00) ( 2.19,  4.45) ( 1.90,  3.73)
              ( 0.12, 0.00) ( 2.88, -3.17)
              (-2.54, 0.00) :End of matrix A

( 3.23, 0.00) ( 1.51, -1.92) ( 1.90,  0.84) ( 0.42,  2.50)
              ( 3.58, 0.00) (-0.23,  1.11) (-1.18,  1.37)
              ( 4.09, 0.00) ( 2.33, -0.14)
              ( 4.29, 0.00) :End of matrix B

```

10.3 Program Results

F08TNF Example Program Results

```

Eigenvalues
-5.9990     -2.9936      0.5047      3.9990

Estimate of reciprocal condition number for B
2.5E-03

Error estimates for the eigenvalues
3.4E-13     2.0E-13      9.6E-14      2.5E-13

```
